Programs for Evaluation of Identified Models of Simulated Data

Söderström, Torsten

1973

Document Version:
Publisher's PDF, also known as Version of record

Link to publication

Citation for published version (APA):
Söderström, T. (1973). *Programs for Evaluation of Identified Models of Simulated Data*. (Techncal Reports TFRT-7052). Department of Automatic Control, Lund Institute of Technology (LTH).

Total number of authors:
1

LUND UNIVERSITY

PO Box 117
221 00 Lund
+46 46-222 00 00

PROGRAMS FOR EVALUATION OF IDENTIFIED MODELS OF SIMULATED DATA

T. SÖDERSTRÖM

Report 7315 (C) July 1973
Lund Institute of Technology
Division of Automatic Control

PROGRAMS FOR EVALUATION OF IDENTIFIED MODELS OF SIMULATED DATA.

T. Söderström

ABSTRACT.

In this report some programs for evaluation of the goodness of models are given. It is generally assumed that the true system is known.

TABLE OF CONTENTS

I. DESCRIPTION OF THE PROGRAMS.

The major part of the programming work has been performed by Mr. E. Burström which is gratefully acknowledged. The routine SLOSS is described in the program library of the Division of Automatic Control. The subroutine PMPY is given in Söderström (1973 d).

Assume that the following system is given:

$$A(q^{-1})y(t) = B(q^{-1})u(t) + C(q^{-1})e(t) \tag{1}$$

$$A(q^{-1}) = 1 + a_1 q^{-1} + \dots + a_{NA} q^{-NA}$$

$$B(q^{-1}) = b_1 q^{-1} + \dots + b_{NB} q^{-NB}$$

$$C(q^{-1}) = 1 + c_1 q^{-1} + \dots + c_{NC} q^{-NC}$$

$e(t)$ white noise $Ee(t) = 0$ $Ee(t)^2 = VE$
$u(t)$ white noise $Eu(t) = 0$ $Eu(t)^2 = VU$
$\{e(t)\}$ and $\{u(t)\}$ independent.

It is also assumed that the following model of the true system is given:

$$\hat{A}(q^{-1})y(t) = \hat{B}(q^{-1})u(t) + \hat{C}(q^{-1})\varepsilon(t) \tag{2}$$

$$\hat{A}(q^{-1}) = 1 + \hat{a}_1 q^{-1} + \dots + \hat{a}_{NMA} q^{-NMA}$$

$$\hat{B}(q^{-1}) = \hat{b}_1 q^{-1} + \dots + \hat{b}_{NMB} q^{-NMB}$$

$$\hat{C}(q^{-1}) = 1 + \hat{c}_1 q^{-1} + \dots + \hat{c}_{NMC} q^{-NMC}$$

The purpose of the programs of this report is to compare the model (2) with the true system (1). In Söderström (1973 c) examples are given in which the programs are used.

1. VML. This program computes the variance of the residuals $\varepsilon(t)$. The inequality

$$E\varepsilon^2(t) \geq VE$$

holds generally, see Söderström (1973 a). If the variance is close to VE the model is good, compare also Söderström (1973 b).

2. MERRO. This program computes the variance of the deterministic model error, i.e. the signal

$$\frac{\hat{B}(q^{-1})}{\hat{A}(q^{-1})} u(t) - \frac{B(q^{-1})}{A(q^{-1})} u(t)$$

For a good model the variance is close to zero. The variance can also be interpreted as the sum of squared deviations of the impulse response, i.e.

$$\sum_{i=1}^{\infty} (\hat{g}_i - g_i)^2$$

where the impulse response of the system (the model) is g_i $(\hat{g}_i)$ at time i after the impulse.

3. MVS. It is assumed that the model (2) is used for the design of a minimal variance regulator. This regulator is assumed to be implemented on the system (1). The variance of the output of the controlled system is computed. Clearly

$$Ey^2(t) \geq VE$$

always holds and if the variance is close to VE the model may be considered as good.

4. PADD. This subroutine performs polynomial addition.

5. POSSIBLE IMPROVEMENTS. It is possible to use a simpler routine for polynomial addition.

II. REFERENCES.

Söderström, T. (1973 a).
On the Uniqueness of Maximum Likelihood Identification for Different Structures. Report 7307, Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 b).
An On-Line Algorithm for Approximate Maximum Likelihood Identification of Linear Dynamic Systems. Report 7308, Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 c).
Programs for Recursive Identification. Report 7312(C), Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 d).
Computations of Asymptotic Least Squares Estimates. Report 7314(C), Division of Automatic Control, Lund Institute of Technology.

III. LISTS OF THE PROGRAMS

```
      SUBROUTINE VML(T,TM,NA,NB,NC,NMA,NMB,NMC,VU,VE,ISTA,V)
C
C     COMPUTES THE VARIANCE OF THE RESIDUALS
C     AUTHOR ERIK BURSTRöM 1972-12-24
C     REVISED TORSTEN SöDERSTRöM 1973-05-01
C     REFERENCE REPORT 7315(C).
C
C     T   -VECTOR OF ORDER (NA+NB+NC) CONTAINING THE SYSTEM PARAMETERS
C     TM  -VECTOR OF ORDER (NMA+NMB+NMC) CONTAINING THE MODEL  PARAMETERS
C     NA   - NUMBER OF A-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NB   - NUMBER OF B-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NC   - NUMBER OF C-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NMA  - NUMBER OF A-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     NMB  - NUMBER OF B-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     NMC  - NUMBER OF C-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     VU THE VARIANCE OF THE INPUT SIGNAL
C     VE THE VARIANCE OF THE NOISE
C     ISTA - IF THE VARIANCE DOES NOT EXIST ISTA =1 (ELSE ISTA=0)
C        AND A MESSAGE IS PRINTED
C     V THE VARIANCE OF THE RESIDUALS
C
C     SUBROUTINE REQUIRED
C        PMPY
C        PADD
C        SLOSS
C
      DIMENSION T(1),TM(1)
      DIMENSION A(12),B(12),C(12),A1(12),B1(12),C1(12),G1(24),G2(24),
     *G3(24),G4(24),G5(24)
      DO 10 I=1,24
      G3(I)=0.
      G4(I)=0.
   10 G5(I)=0.
      ZR=1.
      IDA=NA+1
      IDB=NB+1
      IDC=NC+1
      IDA1=NMA+1
      IDB1=NMB+1
      IDC1=NMC+1
      A(1)=1.
      B(1)=0.
      C(1)=1.
      A1(1)=1.
      B1(1)=0.
      C1(1)=1.
      ISTA1=0.
      ISTA2=0.
      DO 1 I=1,NA
    1 A(I+1)=T(I)
      DO 2 I=1,NB
    2 B(I+1)=T(I+NA)
      DO 3 I=1,NC
    3 C(I+1)=T(I+NA+NB)
      DO 4 I=1,NMA
    4 A1(I+1)=TM(I)
      DO 5 I=1,NMB
    5 B1(I+1)=TM(I+NMA)
      DO 6 I=1,NMC
    6 C1(I+1)=TM(I+NMA+NMB)
C
C
      CALL PMPY(G1,IDG1,A1,IDA1,B,IDB)
```

```
      CALL PMPY(G2,IDG2,A,IDA,B1,IDB1)
      DO 7 I=1,IDG2
    7 G2(I)=-G2(I)
      CALL PADD(G1,G1,G2,G2,G3,G4,ZR,ZR,D,D,IDG1,IDG2,IDG3,1)
      CALL PMPY(G4,IDG4,A,IDA,C1,IDC1)
C
C
      IDGG=MAX0(IDG3,IDG4)-1
      CALL SLOSS(G4,G3,IDGG,ISTA1,V1)
      IF(ISTA1.EQ.0)PRINT 101
  101 FORMAT(/10X,'THE A-POLYNOMIAL HAS ANY ROOT OUTSIDE THE UNIT CIRCLE
     *')
      V1=V1*VU
C
C
      CALL PMPY(G5,IDG5,A1,IDA1,C,IDC)
      CALL PMPY(G4,IDG4,A,IDA,C1,IDC1)
C
C
      IDGH=MAX0(IDG4,IDG5)-1
      CALL SLOSS(G4,G5,IDGH,ISTA2,V2)
      IF(ISTA2.EQ.0)PRINT 101
      V2=V2*VE
      V=V1+V2
      RETURN
      END
```

```
      SUBROUTINE MERRO(T,TM,NA,NB,NMA,NMB,V)
C
C     COMPUTES THE VARIANCE OF THE DETERMINISTIC MODEL ERROR
C     AUTHOR ERIK BURSTRÖM 1972-12-24
C     REVISED TORSTEN SÖDERSTRÖM 1973-05-01
C     REFERENCE REPORT 7315(C).
C
C     T   -VECTOR OF ORDER (NA+NB) CONTAINING THE SYSTEM PARAMETERS
C     TM  -VECTOR OF ORDER (NMA+NMB)  CONTAINING THE MODEL  PARAMETERS
C     NA   - NUMBER OF A-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NB   - NUMBER OF B-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NMA - NUMBER OF A-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     NMB - NUMBER OF B-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     V THE VARIANCE OF THE DETERMINISTIC MODEL ERROR
C
C     SUBROUTINE REQUIRED
C        PMPY
C        PADD
C        SLOSS
C
      DIMENSION T(1),TM(1)
      DIMENSION A(12),B(12),      A1(12),B1(12),G5(24),G1(24),G2(24),
     *G3(24),G4(24)
      DO 10 I=1,24
      G3(I)=0.
   10 G5(I)=0.
      ISTA=0
      ZR=1.
      IDA=NA+1
      IDB=NB+1
      IDA1=NMA+1
      IDB1=NMB+1
      A(1)=1.
      B(1)=0.
      A1(1)=1.
      B1(1)=0.
      DO 1 I=1,NA
    1 A(I+1)=T(I)
      DO 2 I=1,NB
    2 B(I+1)=T(I+NA)
      DO 3 I=1,NMA
    3 A1(I+1)=TM(I)
      DO 4 I=1,NMB
    4 B1(I+1)=TM(I+NMA)
C
C
      CALL PMPY(G1,IDG1,A,IDA,B1,IDB1)
      CALL PMPY(G2,IDG2,A1,IDA1,B,IDB)
      DO 5 I=1,IDG2
    5 G2(I)=-G2(I)
      CALL PADD(G1,G1,G2,G2,G3,G4,ZR,ZR,D,D,IDG1,IDG2,IDG3,1)
      CALL PMPY(G5,IDG5,A,IDA,A1,IDA1)
C
C
      IDGG=MAX0(IDG3,IDG5)-1
      CALL SLOSS(G5,G3,IDGG,ISTA,V)
      IF(ISTA.EQ.0)PRINT 101
  101 FORMAT(/10X,'THE A-POLYNOMIAL HAS ANY ROOT OUTSIDE THE UNIT CIRCLE
     *')
      RETURN
      END
```

```
      SUBROUTINE MVS(T,TM,NA,NB,NC,NMA,NMB,NMC,VE,ISTA,V)
C
C     COMPUTES THE VARIANCE OF THE OUTPUT OF THE SYSTEM, WHEN IT IS
C        CONTROLLED BY A MINIMUM VARIANCE STRATEGY CONSTRUCTED
C        FROM THE MODEL
C     AUTHOR ERIK BURSTRÖM 1972-12-24
C     REVISED TORSTEN SÖDERSTRÖM 1973-05-01
C     REFERENCE REPORT 7315(C).
C
C     T   -VECTOR OF ORDER (NA+NB+NC) CONTAINING THE SYSTEM PARAMETERS
C     TM  -VECTOR OF ORDER (NMA+NMB+NMC) CONTAINING THE MODEL  PARAMETERS
C     NA   - NUMBER OF A-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NB   - NUMBER OF B-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NC   - NUMBER OF C-PARAMETERS OF THE SYSTEM (MIN 0, MAX 10)
C     NMA  - NUMBER OF A-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     NMB  - NUMBER OF B-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     NMC  - NUMBER OF C-PARAMETERS OF THE MODEL  (MIN 0, MAX 10)
C     VE THE VARIANCE OF THE NOISE
C     ISTA - IF THE VARIANCE DOES NOT EXIST ISTA =1 (ELSE ISTA=0)
C        AND A MESSAGE IS PRINTED
C     V THE VARIANCE OF THE OUTPUT
C
C     SUBROUTINE REQUIRED
C        PMPY
C        PADD
C        SLOSS
C
      DIMENSION T(1),TM(1)
      DIMENSION A(12),B(12),C(12),A1(12),B1(12),C1(12),G1(24),G2(24),
     *G3(24),G4(24),G5(24),G6(24),G7(24)
      DO 100 I=1,24
      G1(I)=0.
  100 G6(I)=0.
      ZR=1.
      ISTA=0.
      A(1)=1.
      B(1)=0.
      C(1)=1.
      A1(1)=1.
      B1(1)=0.
      C1(1)=1.
      IDA=NA+1
      IDB=NB+1
      IDC=NC+1
      IDA1=NMA+1
      IDB1=NMB+1
      IDC1=NMC+1
      DO 1 I=1,NA
    1 A(I+1)=T(I)
      DO 2 I=1,NB
    2 B(I+1)=T(I+NA)
      DO 3 I=1,NC
    3 C(I+1)=T(I+NA+NB)
      DO 4 I=1,NMA
    4 A1(I+1)=TM(I)
      DO 5 I=1,NMB
    5 B1(I+1)=TM(I+NMA)
      IF(NMC.EQ.0) GO TO 10
      DO 6 I=1,NMC
    6 C1(I+1)=TM(I+NMA+NMB)
   10 CONTINUE
C
C
```

```
      CALL PMPY(G1,IDG1,B1,IDB1,C,IDC)
      CALL PMPY(G2,IDG2,A,IDA,B1,IDB1)
      CALL PMPY(G3,IDG3,A1,IDA1,B,IDB)
      DO 7 I=1,IDG3
    7 G3(I)=-G3(I)
      CALL PADD(G2,G2,G3,G3,G4,G7,ZR,ZR,D,D,IDG2,IDG3,IDG4,1)
      CALL PMPY(G5,IDG5,B,IDB,C1,IDC1)
      CALL PADD(G4,G4,G5,G5,G6,G7,ZR,ZR,D,D,IDG4,IDG5,IDG6,1)
C
C
      DO 11 I=1,IDG1
   11 G1(I)=G1(I+1)
      IDG1=IDG1-1
      DO 12 I=1,IDG6
   12 G6(I)=G6(I+1)
      IDG6=IDG6-1
      IDG=MAX0(IDG1,IDG6)-1
      CALL SLOSS(G6,G1,IDG,ISTA,V)
      IF(ISTA.EQ.0)PRINT 101
  101 FORMAT(/10X,'THE A-POLYNOMIAL HAS ANY ROOT OUTSIDE THE UNIT CIRCLE
     *')
      IF(ISTA.EQ.0)RETURN
      V=V*VE
      RETURN
      END
```

```
      SUBROUTINE PADD(AR,AI,BR,BI,CR,CI,ZR,ZI,CZR,CZI,IDA,IDB,IDC,IR)
C
C     THIS SUBROUTINE ADDS TWO GIVEN POLYNOMIALS AND COMPUTES THE VALUE
C     OF THE NEW POLYNOMIAL IN A CERTAIN POINT.
C     AUTHOR ERIK BURSTRÖM 1972-12-24
C
C     A - A ARE THE COEFFICIENTS OF THE POLYNOMIAL
C
C         P(S)=A(0)+A(1)*S**1+...+A(NA)*S**NA
C
C         WHERE A(J)=AR(J)+I*AI(J).
C
C     B - B ARE THE COEFFICIENTS OF THE POLYNOMIAL
C
C         P(S)=B(0)+B(1)*S**1+...+B(NB)*S**NB
C
C         WHERE B(J)=BR(J)+I*BI(J).
C
C     C - C ARE THE COEFFICIENTS OF THE POLYNOMIAL
C
C         P(S)=C(0)+C(1)*S**1+...+C(NC)*S**NC
C
C         WHERE C(J)=CR(J)+I*CI(J).
C
C     Z - Z IS THE POINT IN OHICH THE VALUE OF THE NEW POLYNOMIAL IS COM-
C         PUTED WHERE Z=ZR+I*ZI.
C     CZ - CZ IS THE VALUE OF THE NEW POLYNOMIAL IN Z WHERE CZ=CZR+I*CZI.
C     IDA - IDA IS THE DIMENSION OF THE POLYNOMIAL A (MIN 1).
C     IDB - IDB IS THE DIMENSION OF THE POLYNOMIAL B (MIN 1).
C     IDC - IDC IS THE DIMENSION OF THE POLYNOMIAL C.
C     IR - IF IR=1 A AND B ARE CONSIDERED REAL INDEPENDENT OF AI RESP. BI.
C
C     SUBROUTINE REQUIRED
C           NONE
C
      DIMENSION AR(1),AI(1),BR(1),BI(1),CR(1),CI(1)
C
      COMPLEX Z,CZ,DZ
      EPS=1.0E-07
      IDC=MAX0(IDA,IDB)
      I1=MIN0(IDA,IDB)
      I2=I1+1
      CZ=(0.,0.)
      Z=CMPLX(ZR,ZI)
      DO 1 I=1,I1
      CR(I)=0.
      CR(I)=AR(I)+BR(I)
      IF(IR.EQ.1)GO TO 1
      CI(I)=0.
      CI(I)=AI(I)+BI(I)
    1 CONTINUE
      IF(I1.EQ.IDC)GO TO 10
      IF(IDA.LT.IDB)GO TO 5
      DO 3 I=I2,IDC
      CR(I)=0.
      CR(I)=AR(I)
      IF(IR.EQ.1)GO TO 3
      CI(I)=0.
      CI(I)=AI(I)
    3 CONTINUE
      GO TO 10
    5 DO 6 I=I2,IDC
      CR(I)=0.
```

```
      CR(I)=BR(I)
      IF(IR.EQ.1)GO TO 6
      CI(I)=0.
      CI(I)=BI(I)
    6 CONTINUE
   10 CONTINUE
      IF(CABS(Z).LT.EPS)GO TO 12
      DO 11 I=1,IDC
      IF(IR.EQ.1)DZ=CMPLX(CR(I),0.)
      IF(IR.NE.1)DZ=CMPLX(CR(I),CI(I))
   11 CZ=CZ+DZ*Z**(I-1)
      GO TO 13
   12 CZ=CMPLX(CR(1),CI(1))
   13 CONTINUE
      CZR=REAL(CZ)
      CZI=AIMAG(CZ)
      RETURN
      END
```